

[Description](#)[Syntax](#)[Remarks and examples](#)[Quick start](#)[Options for import spss](#)[Stored results](#)[Menu](#)[Option for export spss](#)[Also see](#)

Description

`import spss` reads into memory a version 16 or higher IBM SPSS Statistics (.sav) file or a version 21 or higher compressed IBM SPSS Statistics (.zsav) file. `import spss` can import up to 32,766 variables at one time (up to 2,048 in Stata/BE). If your SPSS file contains more variables than this, you can break up the SPSS file into multiple Stata datasets.

`export spss` writes Stata's data to an IBM SPSS Statistics (.sav) file.

Quick start

Import the IBM SPSS Statistics file `myfile.sav` into Stata

```
import spss myfile
```

Same as above, but replace the data in memory

```
import spss myfile, clear
```

Same as above, but import only variables `x1` and `x4`

```
import spss x1 x4 using myfile, clear
```

Import the compressed IBM SPSS Statistics file `compfile.zsav` into Stata

```
import spss compfile, zsav
```

Same as above, but read variable names as lowercase

```
import spss compfile, zsav case(lower)
```

Export data in memory to `mydata.sav`

```
export spss mydata
```

Same as above, but export only variables `v1` and `v2`

```
export spss v1 v2 using mydata
```

Menu

import spss

File > Import > SPSS data (*.sav)

export spss

File > Export > SPSS data (*.sav)

Syntax

Load an IBM SPSS Statistics file (*.sav)

```
import spss [using] filename [ , options ]
```

Load a compressed IBM SPSS Statistics file (*.zsav)

```
import spss [using] filename , zsav [ options ]
```

Load a subset of an IBM SPSS Statistics file (*.sav)

```
import spss [namelist] [if] [in] using filename [ , options ]
```

Load a subset of a compressed IBM SPSS Statistics file (*.zsav)

```
import spss [namelist] [if] [in] using filename , zsav [ options ]
```

Save data in memory to an IBM SPSS Statistics file (*.sav)

```
export spss [using] filename [if] [in] [ , replace ]
```

Save subset of variables in memory to an IBM SPSS Statistics file (*.sav)

```
export spss [varlist] using filename [if] [in] [ , replace ]
```

If *filename* is specified without an extension, .sav is assumed unless you specify the zsav option, in which case extension .zsav is assumed. If *filename* contains embedded spaces, enclose it in double quotes.

namelist specifies SPSS variable names to be imported.

<i>options</i>	Description
case(<u>l</u> ower <u>u</u> pper <u>p</u> reserve)	read variable names as lowercase or uppercase; the default is to preserve the case
clear	replace data in memory
encoding(" <i>encoding</i> ")	specify the file encoding; see help encodings

collect is allowed with import spss; see [\[U\] 11.1.10 Prefix commands](#).

encoding() does not appear in the dialog box.

Options for import spss

`zsav` indicates the file to load is a compressed IBM SPSS Statistics file.

`case(lower | upper | preserve)` specifies the case of the variable names after import. The default is `case(preserve)`.

`clear` specifies that it is okay to replace the data in memory, even though the current data have not been saved to disk.

The following option is available with `import spss` but is not shown in the dialog box:

`encoding("encoding")` specifies the encoding of the file. If your file has an incorrect encoding specified in the file header, you can use this option to specify the correct encoding. See [help encodings](#) for details.

Option for export spss

`replace` specifies that *filename* be replaced if it already exists.

Remarks and examples

`import spss` reads into memory a version 16 or higher IBM SPSS Statistics (`.sav`) file or a version 21 or higher compressed IBM SPSS Statistics (`.zsav`) file. If an SPSS variable name from the file does not conform to a Stata variable name, a generic `v#` name will be assigned, and the original variable name will be stored as a characteristic for the variable. If an SPSS variable label is too long, it will be truncated to 80 characters, and the original variable label will be stored as a variable characteristic. All value labels for string variables will be ignored. Value labels for numeric variables will be named `labels#` and attached to the corresponding variable.

`export spss` exports the Stata dataset in memory, or a subset of the dataset, to an IBM SPSS Statistics (`.sav`) file.

► Example 1: Importing an SPSS file into Stata

We can import SPSS files into Stata, either by selecting the entire file or by selecting subsets of the data, with `import spss`. For example, we have the SPSS file `auto.sav`, which contains data on automobiles. Below, we demonstrate how to import these data into Stata. To follow along, download this file to your working directory by typing the copy command below:

```
. copy https://www.stata.com/sampledatab/auto.sav auto.sav
```

We first load the entire `auto.sav` file into Stata by typing

```
. import spss auto
(12 vars, 74 obs)
```

```
. describe
```

Contains data

Observations: 74

Variables: 12

Variable name	Storage type	Display format	Value label	Variable label
<hr/>				
make	str17	%17s		
price	int	%5.0f		
mpg	byte	%2.0f		
rep78	byte	%1.0f		
headroom	double	%3.1f		
trunk	byte	%2.0f		
weight	int	%4.0f		
length	int	%3.0f		
turn	byte	%2.0f		
displacement	int	%3.0f		
gear_ratio	double	%4.2f		
foreign	byte	%1.0f		

Sorted by:

Note: Dataset has changed since last saved.

We can instead import only variables `make` and `weight` into memory from `auto.sav`. We include the `clear` option to replace the data in memory without saving them.

```
. import spss make weight using auto, clear
(2 vars, 74 obs)
```

```
. describe
```

Contains data

```
Observations:      74
Variables:         2
```

Variable name	Storage type	Display format	Value label	Variable label
make	str17	%17s		
weight	int	%4.0f		

```
Sorted by:
Note: Dataset has changed since last saved.
```



► Example 2: Export all Stata data to an SPSS file

We have a data on food consumption. The data is currently in the Stata `.dta` format. We would like to export the data to an IBM SPSS Statistics file. We first load the Stata dataset and then export the data to a file named `food_consumption.sav`:

```
. use https://www.stata-press.com/data/r19/food_consumption, clear
(Food consumption)

. export spss food_consumption
file food_consumption.sav saved
```



Stored results

`import spss` stores the following in `r()`:

Scalars	
<code>r(N)</code>	number of observations imported
<code>r(k)</code>	number of variables imported

Also see

- [D] **import** — Overview of importing data into Stata
- [D] **export** — Overview of exporting data from Stata

